Historicamente, o mIRC sempre teve a possibilidade de definir comandos adicionais,


     além dos que já vêm incluídos "de fábrica". Essa característica tornou possível a criação


     de muitos scripts padronizados (tais como o FireFight, C-Script, 7thSphere e muitos


     outros), mas não é conhecida de muitos usuários, que se limitam a usar scripts prontos


     sem se preocupar em descobrir como eles são feitos. O help do mIRC é muito completo e


     fácil de entender; todavia, uma questão cultural e lingüística impede grande parte dos


     usuários brasileiros de tentarem compreendê-lo - em parte devido à conhecida resistência


     dos brasileiros a lerem qualquer tipo de manual de instruções ;-) , e também pelo fato de o


     help ser em inglês. 





         Pois bem, os scripts são separados em partes: Aliases, Popups, Remote Commands,


     Remote Events, Remote Raw Events e DDE Extensions... 











                                                          





                                           





                                                          





                                                           





         Os aliases, juntamente com os Popups, Remote Commands, Remote Events, Remote


     Raw Events e DDE Extensions, formam o esqueleto de qualquer script, e constituem uma


     série de definições de novos comandos, que podem ser executados diretamente da linha de


     comando, ou então chamados por outro alias, popup ou remote. 





         Para editar os seus aliases, use a opção Tools / Aliases. O mIRC abrirá o seu editor


     de scripts, e provavelmente exibirá uma série de aliases já pré-definidos (pelo próprio mIRC


     ou pelo script que você estiver usando). 








                            Identificadores





              Identificadores Básicos








         $date - data atual


         $time - hora atual


         $idle - idle time (mesmo do /whois)


         $asc(V) - retorna o código ASCII do caracter


         V


         $chr(n) - retorna o caracter cujo código ascii


         é n


         $rand(v1,v2) - retorna um número aleatório


         $me - retorna o seu nick atual


         $away - indica se você está away ou não


         $cb - primeiros 256 caracteres da clipboard


         $ip - seu endereço ip


         $server - o servidor no qual você está


         conectado


         $port - a porta que você está usando





                                                    Funções








                                             Strings: $abs, $asc, $chr,


                                             $count, $left, $len, $lower,


                                             $mid, $pos, $remove,


                                             $replace, $right, $str, $strip,


                                             $upper





                                             Arquivos & Diretórios: $dir,


                                             $file, $hfile, $exists, $findfile,


                                             $getdir, $lines, $lof, $logdir,


                                             $mididir, $nofile, $nopath,


                                             $mircdir, $mircini, $read,


                                             $readini, $wavedir





                                             Nicks e endereços: $adress,


                                             $comchan, $ial, $level, $mask,


                                             $me, $nick, $nopnick, $notify,


                                             $nvnick, $opnick, $nvnick,


                                             $opnick, $snicks, $snick,


                                             $vnick





                                             Janela: $active, $chan, $chat,


                                             $fserv, $get, $query, $send





                                             Tokens: $addtok, $findtok,


                                             $gettok, $instok, $remtok,


                                             $reptok














                        FORMATO BÁSICO











    Cada alias é definido em (no mínimo) uma linha, que indica o nome do comando que você deseja


criar e a ação que deve ser executada. Veja um alias simples :





/oi /echo 4 -s Oi!!!





    Se você inserir este alias e pressionar o botão Ok, terá criado um comando /Oi, que nada mais


faz além de mostrar a palavra " Oi!!!" em letras vermelhas na sua janela de status. Teste-o e maravilhe-se


;)





    Opcionalmente, você pode remover as barras dos comandos dos aliases. Um alias assim :





oi echo 4 -s Oi!!!





    Funcionaria de maneira idêntica ao anterior. Mas... esse alias está muito simples. O que


aconteceria se eu precisasse utilizar mais de um comando do mIRC? É muito simples... Vamos definir


um alias que diga duas palavras, "Oi" e "Tchau", cada uma em uma linha diferente da sua janela de


status :





oitchau {


echo 4 -s Oi!!!


echo 4 -s Tchau!!!


}





     Viu que simples? A chave "{" logo após o nome do alias indica ao mIRC que todos os comandos


     indicados nas linhas seguintes devem ser executados, até encontrar a chave "}" correspondente.








                        IDENTIFICADORES











    Identificadores são palavras reservadas pelo mIRC para uso em scripts, que retornam dados


importantes sobre a configuração ou as condições do seu mIRC, canais, janelas e usuários. 





     Uma lista completa de identificadores pode ser lida digitando o comando /HELP IDENTIFIERS, e


     os identificadores mais importantes estão assinalados no Quadro 1. Vamos a alguns exemplos do


uso de identificadores.





Listando a configuração





    Digamos que você queira fazer um alias que informe a configuração atual do mIRC do seu


usuário. Uma possibilidade seria assim : 





/Status_all {


/echo 4 -s Status global 


/echo 3 -s Seu nick é $me e seu ip é $ip 


/echo 3 -s O relógio de seu micro marca $time


/echo 3 -s Você está conectado na porta $port do $server 


/echo 3 -s Seu modo é $usermode 


/echo 3 -s Download = $getdir - Logs = $logdir - mIRC = $mircdir


/echo 3 -s Você está em $chan(0) canais 


/echo 3 -s e tem $query(0) pvts e $chat(0) DCC Chats abertos. 


/echo 3 -s A janela ativa é $active


} 





    Este alias nada mais faz do que mostrar na sua janela de status uma série de informações sobre


sua configuração.








                    USANDO PARÂMETROS








    Vamos a um exemplo menos simples para poder demonstrar bem o poder dos identificadores.


Para isto, precisaremos do conceito de parâmetros. Parâmetros são as palavras inseridas pelo usuário


na linha de comando, sempre que ele vai executar o seu alias. Por exemplo, se um usuário digita :


/multijoin #cIRCo #Floripa #Brasircop as palavras "#cIRCo", "#Floripa", e "#Brasircop" são os parâmetros. O


seu alias sempre tem como saber os parâmetros que o usuário digitou, utilizando-se de identificadores


especiais : os identificadores de parâmetros. Os identificadores de parâmetros são chamados utilizando


o sinal de "$" e o número do parâmetro desejado, assim temos : $1 é o primeiro parâmetro, $2 é o


segundo parâmetro, e por aí afora.





    Podemos complicar um pouco, com os identificadores de múltiplos parâmetros. Por exemplo, $1-3


indica os parâmetros de 1 a 3. $2-5 indica os parâmetros de 2 a 5. $1- indica todos os parâmetros da


linha. $3- indica todos os parâmetros a partir do terceiro. 





    Vamos a um exemplo?





/mostra_params {


/echo -s os parâmetros digitados foram $1-


/echo a terceira palavra é $3


/echo as primeiras três palavras são $1-3


}


    O exemplo acima simplesmente mostra os parâmetros digitados. Experimete digitar :





/mostra_params Estou aprendendo a programar o mIRC





    Para ter uma idéia de como ele funciona. Agora vamos ao /multijoin





/multijoin /join $1 $+ , $+ $2 $+ , $+ $3





    Este alias faz com que você entre em até 3 canais com um único comando. Por exemplo, se você


digitar assim : 





/multijoin #cIRCo #floripa #amizade





    o mIRC converterá este comando para /join #cIRCo,#floripa,#amizade





    veja bem : sem espaços entre as vírgulas e os nomes dos canais. Quem suprime estes espaços é


o identificador $+ , responsável por unir dois textos sem deixar espaços extras entre eles.





Contando usuários em um canal





    Um exemplo bastante prático de script é o censo de canal - aquelas mensagens que muita gente


mostra, dizendo quantos usuários estão em um canal, e quantos são operadores do mesmo. Que tal


colocar uma mensagem assim no seu mIRC? É simples...





censo {


/msg $$1 Existem $nick(0,$$1) usuários conectados ao canal $$1 ,


/msg $$1 dos quais $opnick(0,$$1) são operadores do canal e 1 é genial : $me 


}





    Este alias usa as funções $nick e $opnick para contar os usuários do canal passado como


parâmetro. Notou que ao invés de usar $1 eu usei $$1? Isso tem uma importância fundamental : quando


se usa o duplo-cifrão, o mIRC recusa-se a executar o alias caso não seja passado nenhum parâmetro.


Experimente entrar no #brasil e digitar :





/censo #brasil





para ver como funciona...








                              VARIÁVEIS








    Variáveis são essenciais em qualquer linguagem de programação séria, e o mIRC script também


as têm. Se você tem boas noções de programação, pode fazer scripts aproveitando todo o seu


potencial. As variáveis do mIRC têm o prefixo %, e você pode defini-las através do comando /set:





/set %contakick 1





    A linha acima cria uma variável ContaKick, e atribui a ela o valor inicial de 1.





Você pode executar diversas operações com variáveis numéricais, tais como :





%x = 5 + 1





%x = 5 - %y





%x = %x * 2





%x = %z / $2





%x = $1 % 3





%x = 2 ^ %w





inc %x 3





dec %x 3








    Vamos fazer um alias de kick que utiliza a variável %ContaKick que nós acabamos de criar, para


manter uma contagem de vítimas de kicks...





/kc {


/kick # $1 Sai fora! ( vítima %contakick )


inc %contakick 1


}








    Agora, sempre que você kickar utilizando o alias /kc, o kick vai incluir uma contagem de vítimas :-)





    O mIRC aceita também variáveis alfanuméricas, mas devido à complexidade do tema, vou deixar


você se virar sozinho nessa >;)








           REDEFININDO TECLAS DE FUNÇÃO








    Usando os aliases, você pode redefinir as teclas de função do seu mIRC, colocando qualquer


função nelas. É muito simples, faça assim :





/F3 /echo -s você pressionou a tecla F3


/sF3 /echo -s você pressionou shift-F3


/cF3 /echo -s você pressionou control-F3 





    Assim, você pode colocar suas funções mais utilizadas em teclas de função,para poder ter acesso


mais rápido a elas. 





                 REPETIÇÕES E CONDIÇÕES








    Já que entramos no terreno da programação séria, temos que abordar o tema das repetições e


condições. Sim, o mIRC também apresenta a estrutura condicional (If-then-else), e é capaz de fazer


repetições simples (goto).





    Novamente, não vou abordar os conceitos, apenas explicar as técnicas para quem já tem noções


de programação. O mIRC implementa a estrutura if-then-else completa, com uma sintaxe bastante


simples, semelhante à de outras linguagens de alto nível. A sintaxe básica é :





if (v1 operador v2) { comandos }


elseif (v1 operador v2) { comandos }


else { comandos }





    Os operadores disponíveis estão no quadro abaixo.





    Para repetição, o mIRC implementa apenas o comando goto. Ele é usado para transferir a


execução de um alias para um determinado label, definido no formato :NomeDoLabel.





    Geralmente você irá utilizar o goto para implementar loops (estruturas repetitivas). A parte mais


importante de um loop é a sua variável de controle (no exemplo, eu usei %i como variável de controle).


Esta variável normalmente é incrementada a cada "volta" do loop, e ao final do mesmo uma instrução IF


verifica se deve continuar no loop ou encerrar.





    Veja o exemplo a seguir, retirado do help do mIRC, onde :next e :done são labels :





listops {


echo 4 * Listando Ops no #


set %i 1


:next


set %nick $nick(%i,#)


if %nick == $null goto done


if %nick isop # echo 3 %nick é um Op!


inc %i


goto next


:done


echo 4 * Fim da lista de Ops


}





    O alias faz um loop entre o :next e o goto next, verificando se cada usuário do canal é um op.


Esta é uma clássica emulação de um loop repeat-until usando goto ;-)








                      OPERADORES DO IF 





             ==             igual a


             !=              diferente de


             <               menor que


             >               maior que


             >=<=         maior ou igualmenor ou igual


             //\\             é múltiplo denão é múltiplo de


             isin            string v1 está em string v2


             iswm          wildstring (*,?) combina com v2


             isnum         v1 é número e está na faixa v2 (n1-n2) (v2 é opcional)


             ison           nick v1 está no canal v2


             isop           nick v1 é op no canal v2


             isvo            nick v1 tem voice no canal v2


             ischan        v1 é um canal no qual você está


             isauto         v1 está na sua auto-op list para canal v2 (v2 é opcional)


             isignore      v1 está na sua ignore list com o switch v2(v2 é opcional)


             isprotect     v1 está na sua protect list para o canl v2 (v2 opcional)


             isnotify       v1 está na sua notify list


             &&             and lógico


             ||                or lógico














                                                          





                                           





                                                          





                                                           








                  Alterando os menus





              Como usar a seção Popups para


                                adaptar 


              os menus às suas necessidades














    O mIRC permite que o usuário altere vários de seus menus popup. As alterações podem ser feitas


nos seguintes menus : 





    janela de status 


    query/chat/notify 


    canal 


    lista de nicks do canal 


    menu principal 





    Estes menus são acessados clicando com o botão direito do mouse nas respectivas áreas. Seu


uso mais comum é para tarefas corriqueiras como dar op, kickar um usuário ou entrar em um canal.





Sintaxe





    Uma definição de linha de menu tem a seguinte sintaxe :





opção : comando





    Exemplo : Kickar usuário: kick # $1 Você foi kickado :)





    A parte que vem antes do ":" é o que aparece no menu - a parte que vem depois do ":" é o


comando a ser executado se esta opção for selecionada. A sintaxe dos comandos é eatamente a


mesma dos aliases, aceitando inclusive os caracteres { } e |





    Você pode criar uma hierarquia de submenus utilizando o símbolo '.' e separar grupos dentro do


menu usando o símbolo '-'





    Exemplo :





Kick


.Não é bem vindo:kick # $1 Você não é bem vindo aqui


.Sai fora mané:kick # $1 Sai fora mané


.-


.Kick + ban


..Eterno: /ban # $1 2


..30 segundos: /ban -u30 # $1 2





                   EDITANDO SEUS MENUS 








    A edição dos menus ocorre na opção Tools|Popups do seu mIRC. Uma vez dentro desta opção,


selecione na opção View qual dos menus você deseja editar. Status é o popup da janela Status. Channel é


o popup da área de textos dos canais. Query/Chat é o popup dos PVTs, Chats e notify window. Nickname list


é o menu da lista de nicks dos canais, e Menubar é o menu dropdown, criado entre as opções Dcc e


Window do menu principal.





Usuários mais avançados podem querer colocar seus menus na área Remote do mIRC. Isto é possível,


colocando o prefixo menu no início da definição.





Exemplo:





menu nicklist {


torno de 660 Kg


.Haddock:/me bate em $1 coim um torno de 600 kg


}








    Os nomes dos menus, para efeito de definição no remote, são : nicklist, channel, query, nicklist e


menubar.





          MENUS DE JANELAS CRIADAS PELO


                               USUÁRIO 








    No remote você também pode criar popups para janelas do tipo criado com o comando /window.


Por exemplo, se você tem uma janela chamada @teste, poderia criar um menu assim :





menu @teste {


[dclick /comando $1]


...Items...


}





    A linha dclick indica o comando a ser executado em caso de double-click na @teste. Neste caso, o


identificador $1 contém o número da linha que foi clicada. 








                                                          





                                           





                                                          





                                                           








                  AUTOMATIZANDO O mIRC 








    Como fazer com que o mIRC reaja automaticamente a diversos eventos





    Aprenda a dominar a seção Remote do seu mIRC, e faça-o executar diversas tarefas sozinho,


para não precisar mais tirar os olhos do canal ;)





    A seção Remote permite criar manipuladores que reagem a eventos diversos, tais como : 





    um usuário entrar no canal 


    kicks & bans 


    sua conexão encerrar 





    Esta é a parte mais complexa de qualquer script de mIRC, e para utilizá-la bem, você precisa já ter


conhecimento de como usar os comandos básicos do IRC, bem como criar aliases e utilizar as


variáveis e identificadores.





    O remote consiste em 3 seções distintas : 





    · Users : define níveis de acesso dos usuários 


    · Variables : mantém a lista das variáveis definidas 


    · Scripts : é o corpo do remote. Aqui são criados as rotinas que respondem aos eventos. Esta


    seção substitui os antigos Commamds, Events e Raw, característicos do mIRC até a versão 4.72.


    Você pode ter vários scripts carregados simultaneamente, o que permite que um mesmo evento do


    sistema acione mais de uma rotina no seu mIRC. 





                       NÍVEIS DE ACESSO 








    Os níveis de acesso são atribuídos tanto a usuários quanto a eventos, e permitem limitar quais


eventos podem ser acessados por cada classe de usuário. Por exemplo, você poderia criar três níveis


de acesso : um para amigos, um para inimigos, e outro para desconhecidos.





    Então, você poderá criar um script para o evento JOIN (entrada de um usuário no seu canal), e,


dependendo do nível de acesso da pessoa que entrar, esse seu script daria op (para os amigos),


kickaria (os inimigos) ou não faria nada (para os desconhecidos).





    Complicado? Calma, que a gente explica.





    O nível de acesso padrão é 1, ou seja, usuários não listados em sua seção de Users recebem do


mIRC o nível 1. Isso significa que qualquer usuário pode ter acesso aos eventos de nível 1. Quanto


maior o nível de acesso de um usuário, mais eventos ele pode acessar.





    Você pode mudar o modo padrão na configuração do mIRC, para permitir que usuários não


listados tenham acesso a mais comandos.





            NíVEIS DE ACESSO DE USUÁRIOS 








    Na seção Users você pode especificar uma lista de usuários e seus níveis de acesso, da seguinte


forma :





[nível1,nível2,...,nívelN]:[usuário]





3,4,8:Bagre!Renato@*.matrix.com.br





    O primeiro nível é o nível de acesso geral, o que significa que o usuário pode acessar todos os


eventos de nível menor ou igual a 3. Todos os outros níveis são níveis específicos, que exigem que o


evento seja exatamente igual para permitir o acesso. Dessa forma, o usuário tem acesso aos eventos de


nível 1, 2, 3, 4 e 8.





    Se você deseja forçar que o primeiro nível de acesso seja para um nível específico, ao invés de


um nível geral, você pode colocar um sinal de igual na frente dele.





=3,4,8:Bagre!Renato@*.matrix.com.br





    Assim, este usuário terá acesso apenas aos eventos de nível 3, 4 e 8, e a nenhum mais. 





           NÍVEIS DE ACESSOS DE EVENTOS 








    O formato geral de eventos é :





[prefixo] [nível]:[evento]:[comandos]





ctcp 1:HELP:/msg $nick Usuários nível 1 não tem acesso ao help





    O evento CTCP acima pode ser acessado por qualquer usuário porque é um evento nível 1.


Assim, se um usuário qualquer mandar a você um /CTCP seunick HELP, seu script enviará a ele a


resposta acima.





    Quando um mesmo usuário provoca resposta em vários níveis de evento, só a de nível mais alto é


executada.





                        NÍVEIS COM NOME








    Você também pode dar nome aos níveis, ao invés de numerá-los. Um nível nomeado age da


mesma maneira que um nível específico, mas são mais fáceis de entender e lembrar.





amigo:bagre!Renato@*.matrix.com.br





on @amigo:JOIN:#Cardume:/mode $chan +o $nick





    Isto define a palavra amigo como um nível de acesso específico. Se um usuário com o nível


"amigo" entrar no canal #Cardume, ele ganhará op.





                      LIMITANDO ACESSO 








    Você pode limitar o acesso de usuários a um evento prefixando-o com um sinal de +.





10:Bagre!Renato@*.matrix.com.br





ctcp +5:HELP:/msg $nick Não há help para usuários nível 5.





    No exemplo acima, o usuário bagre não poderá acessar o evento, porque, mesmo tendo nível de


acesso superior ao do evento, o evento está limitado apenas a usuários com nível exatamente igual a 5.





    Você pode evitar que usuários de maior nível acessem eventos de menor nível usando um sinal de


=.





10:Bagre!Renato@*.matrix.com.br





ctcp 2:HELP:/msg $nick Não há help para usuários nível 2





ctcp 5:HELP:=





    O usuário Bagre não poderá acessar nenhum destes eventos porque o evento de nível 5 impede-o


de acessar todos os eventos CTCP HELP de níveis inferiores a 5.





    Você pode impedir um evento de reagir se ele foi ativado por você mesmo, utilizando o ponto de


exclamação.





ctcp !2:HELP:/msg $nick Não há help para usuários de nível 2





    Você não poderia utilizar este evento, independente do nível de acesso que você tenha.





    Você pode limitar eventos para serem executados apenas quando você é op no canal, usando a


arroba.





10:Bagre!Renato@*.matrix.com.br





on @2:JOIN:#Cardume:/mode $chan +o $nick





    Quando o usuário acima entrar no #Cardume e você estiver como op lá, o comando /mode


associado será executado, e dará op ao usuário. Se você não for op, o evento não será executado.





    Você pode evitar que eventos de um certo nível sejam executados, usando um ponto de


exclamação no lugar do comando.





ctcp 5:PING:echo PING! ctcp 5:*:! 





    O ! no fim da linha informa ao mIRC para encerrar qualquer processamento posterior de


comandos ctcp nível 5.








                       CRIANDO GRUPOS 








    Você pode criar grupos dentro dos scripts usando o sinal #.





#grupo1 on





...





[ lista de eventos ]





...





#grupo1 end





    Você pode usar os comandos /enable e /disable para habilitar ou desabilitar um grupo. Um grupo


desabilitado será ignorado na hora de processar eventos. Ah, você não pode ter grupos dentro de


grupos.








                  ORDEM DAS DEFINIÇÕES 








    Muitos dos prefixos e controles são sensíveis à ordem numérica das definições. O mais seguro é


ordenar suas definições começando com os níveis de acesso mais baixo e ir subindo, para facilitar as


coisas tanto para você mesmo quanto para o mIRC ;)





                   COMANDOS DO REMOTE








    Alguns comandos específicos do mIRC alteram configurações das seções do Remote. Vou


descrevê-los aqui, mas recomendo que você consulte o help do mIRC para compreendê-los melhor.








                       COMANDOS GERAIS








    Os comandos /ctcps, /events e /raw ligam (ON) ou desligam (OFF) o processamento de cada um


dos tipos de remote. 





    O comando /remote liga ou desliga o processamento de todos os eventos.





    O comando /dlevel nível altera o nível padrão para usuários não listados na seção users.





    O comando /groups lista os grupos definidos em seu script. Se você digitar /groups -e, listará


apenas os grupos ativos. Se você digitar /groups -d, listará os grupos desabilitados.





    O comando /enable ativa um (ou mais) grupos. O comando /disable os desativa.





           COMANDOS DE USUÁRIOS E NÍVEIS








    O comando /auser adiciona um usuário à sua lista, ou adiciona níveis a um usuário já existente.





    Se você digitar /auser 1,2,3 Bagre, será criado um usuário Bagre com os acessos 1, 2 e 3. 





    Se você digitar /auser -a 4,6 bagre, serão adicionados os níveis 4 e 6 ao usuário bagre, caso este


já exista. Caso ele não exista, será criado com estes níveis. 





    Veja também o help do comando /guser para saber como criar os usuários já com hostmasks.





    O comando /flush limpa a sua lista de usuários. Se você digitar /flush 1,2,3, o mIRC irá remover


todos os usuários de níveis 1, 2 e 3 que não estejam neste momento nos mesmos canais onde você


está.





    Veja também o comando /ruser, que remove níveis de usuários ou os registros completos dos


usuários, e o comando /ulist, que lista usuários de um determinado nível.





    O mIRC tem uma série de identificadores que podem ser usados nos seus scripts de remote.





              IDENTIFICADORES DO REMOTE





         $adress                     Retorna o endereço do usuário associado ao evento,


                                           no formato login@máquina.domínio


         $chan                        O nome do canal associado ao evento


         $clevel                      O nível associado ao evento


         $dlevel                      Nível default de usuário


         $fulladress                Retorna o endereço completo do usuário associado


         ao 


                                           evento (nick!login@maquina.domínio)


         $group                       Retorna o nome ou status de um dado grupo.


         $maddress(nick!userid@address)          O primeiro endereço na lista de 


                                           usuários que combina com o endereço dado.


         $nick                         O nick do usuário associado ao evento.


         $numeric                   O código de um evento Raw


         $script                       O nome do script sendo executado


         $scriptdir                  O diretório do script sendo executado


         $site                          A parte do $adress depois do @


         $target                      O target de um evento


         $ulevel                      O nível de usuário atingido por um evento


         $wildsite                   *!*@ seguido por $site














                          EVENTOS CTCP








    Os eventos CTCP formam uma maneira especial de comunicar-se dentro do IRC. A definição de


um evento CTCP é feita na seção Scrips , com o seguinte formato :





ctcp [nível]:[texto]:[comandos] 





    O nível é o nível necessário para acessar este evento, o texto é o comando CTCP que irá acionar


este evento, e os comandos são a lista de comandos a ser executada quando este evento for acionado.





    Exemplo : dando op aos amigos





=5:bagre!Renato@*.matrix.com.br 





ctcp 5:opme:/mode $1 +o $nick





ctcp 5:inviteme:/invite $nick $1 





    Estas definições permitem que o usuário Bagre a solicitarem que o seu mIRC dê op ou invite a


ele, utilizando-se de comandos CTCP para isso. Assim, se Bagre digitasse : /ctcp SeuNick opme


#Cardume  e você estivesse como op no #Cardume, automaticamente você daria op ao Bagre.








               ALTERANDO UMA RESPOSTA


                        PADRÃO DE CTCP








    Alguns comandos CTCP são pré-definidos pelo mIRC. Alguns exemplos são o PING, USERINFO,


CLIENTINFO, VERSION e TIME. Eles tem respostas padrão, mas você pode alterá-las facilmente, como


qualquer outro CTCP.





ctcp 1:ping:/notice $nick PONG!!!! | /halt 





    Este script fará que, sempre que alguém lhe enviar um PING, você responda automaticamente


com um "PONG!!!!". O /halt no fim da linha impede que a resposta padrão de ping seja enviada - caso


você retire o /halt, a resposta padrão é enviada juntamente com a sua resposta.





    O único CTCP que não pode ter sua resposta padrão suprimida é o VERSION, porque o mIRC


gosta de propaganda ;)








                    CONTROLANDO O mIRC


                          REMOTAMENTE








    Uma das formas de controlar o mIRC remotamente é via CTCP (embora não seja a melhor).


Digamos que você esteja utilizando clones, e deseja controlar todos eles a partir de um único mIRC.


Defina um usuário para você nos clones "escravos", e dê a ele (ou seja, a você mesmo) um nível de


acesso superior a todos os outros.





100:*!acidwar@*.acidwar.com 





    E defina os comandos que você deseja executar remotamente. Por exemplo, eis a linha que


implementa um comando quit.





ctcp 100:desconecte:{





/notice $nick Recebido - encerrando atividades 





/quit





} 





    Esta próxima linha implementa um comando que ordena que o "escravo" envie um determinado


arquivo a alguém :





ctcp 100:send:/dcc send $1 $2- 





    Com estas duas definições, você pode ordenar que seu "escravo" transfira arquivos, assim : 





/ctcp NickEscravo send Bagre tarefas.txt 





    O que faria com que o escravo enviasse ao Bagre o arquivo tarefas.txt.





    Quando cansar de enviar arquivos com o escravo, você pode digitar : /ctcp NickEscravo


desconecte e o seu escravo se desconectará do servidor.





    Tome muito cuidado com os comandos que você define, pois você pode acabar permitindo que


pessoas não autorizadas façam coisas potencialmente perigosas no seu computador. Da mesma forma,


cuidado ao aceitar scripts de estranhos - você nunca sabe que comandos remotos eles podem ter


embutido no seu script.








                 "WILDCARDS" E VARIÁVEIS








    Você pode usar wildcards (* e ?) na definição dos seus CTCPs. Por exemplo :





ctcp 1:*help*:/notice $nick Você precisa de ajuda?





    Com o comando acima, você responderá automaticamente a qualquer CTCP que contenha a


palavra HELP dentro de si. Exemplos :





/ctcp SeuNick HELP





/ctcp SeuNick NEEDHELP





/ctcp SeuNick HELPME





    Você também pode usar variáveis. Assim , se você definir o seguinte script :





ctcp 1:%senha:/notice $nick Acesso autorizado





    Este evento só será executado caso algum usuário envie o conteúdo correto da variável %senha.


Assim, se você digitar...





/set %senha bigmac ...o evento acima só funcionará se algum usuário digitar /ctcp SeuNick bigmac.








                           EVENTOS RAW








    Os raw remotes são a maneira mais poderosa (e mais complicada) de alterar o comportamento do


seu mIRC. Eles simplesmente redefinem a maneira como o mIRC reage a eventos padrão, tais como : o


resultado de um /list, o resultado de um /links, um /whois, e muitos outros.





    A alteração no raw só pode ser feita conhecendo os códigos numéricos das mensagens do


server, definidos na RFC1459 (Em Inglês). Scripts poderosos usam raw para alterar profundamente a


aparência e comportamento do mIRC. 





    Um exemplo de uso do Raw é a ferramenta "Procurar IRCops" que vem no DusK 5.0 e


posteriores. Caso você tenha interesse em alterar o raw, leia a RFC1459 com atenção e veja exemplos


em outros scripts.





    Um exemplo simples de Raw é o seguinte (adaptado do help do mIRC) :





raw 322:*floripa*:/echo 5 $1-





    Esta linha altera a resposta do comando /list, mostrando de maneira especial todos os canais que


contenham a palavra floripa em seu nome ou topic.








                        EVENTOS PADRÃO








    Os eventos padrão são eventos pré-definidos pelo mIRC, para alterar a resposta padrão à maior


parte dos acontecimentos comuns do mIRC. Seu formato geral é:





on [nível>:[nome]:[condições]:[comandos]





    Por exemplo : o evento ON JOIN é ativado sempre que um usuário entra em um canal no qual você


esteja, e pode ser definido assim :





on [nível]:JOIN:[#[,#]]:[comandos]





    Se você definir um script assim :





on 1:JOIN:#brasil,#turma:/notice $nick Olá, bem vindo ao $chan





    Você passará a mandar uma mensagem a todos os usuários que entrarem no canal #brasil ou


#turma, desde que você também esteja nestes canais. Cada um dos eventos padrão tem sintaxes e


características próprias. Vou citar aqui os eventos padrão do mIRC, mas deixarei a você a tarefa de


pesquisar no help a sintaxe de cada um deles.





              IDENTIFICADORES DO REMOTE





             Action                   alguém manda um /me para um canal ou pvt


             Ban                       alguém é banido em um canal


             Chat                      uma mensagem em DCC Chat


             ChatClose             um DCC Chat é encerrado


             ChatOpen             um DCC Chat é aberto


             Connect                Você se conecta ao server


             CTCPReply           Você recebe uma resposta padrão de CTCP


             Deop                     Alguém perde op em um canal


             Devoice                Alguém perde voice em um canal


             DNS                      Fim de um comando /dns


             Error                     Server manda um ERROR


             FileRcvd               Fim bem-sucedido de um DCC GET


             FileSent               Fim bem-sucedido de um DCC SEND


             GetFail                 Erro em DCC Get


             Input                    Entrada de dados em uma textbox


             Invite                   Recepção de um invite


             Join                     Usuário entra em um canal


             Kick                     Usuário é kickado em um canal


             Load                    Script é carregado


             MidiEnd               Música MIDI termina


             Mode                   Modo do canal é alterado


             Nick                     Alguém muda de nick


             NoSound             Alguém manda um /sound e você não tem


             Notice                 Recepção de um notice














                                                          





                                           





                                                          





                                                           





                      PROTEÇÃO PESSOAL








    A proteção pessoal geralmente diz respeito a defesa contra CTCP flood. O CTCP flood é uma das


maiores armas disponíveis para os conflitos via IRC. Ele opera basicamente da seguinte forma : o seu


mIRC vem configurado para responder automaticamente a uma série de mensagens especiais (as


mensagens CTCP) sem que você faça nada para interferir, e sem que você perceba, a não ser que


você deixe sua janela Status permanentemente aberta (eu deixo).





    Utilizando-se de algumas técnicas bastante comuns, um usuário pode, através do envio de muitas


mensagens CTCP (tais como : PING, VERSION, USERINFO e CLIENTINFO), forçar o seu mIRC a


enviar um excesso de mensagens de resposta, o que faria com que o servidor ao qual você está


conectado detectasse que você está fazendo flood (e tecnicamente você está, embora você não tenha


culpa) e o desconectasse imediatamente da rede, com uma mensagem semelhante a esta : 





                           *** User1 has quit IRC (Excess flood) 





    O script que vamos abordar utiliza o seguinte método para se defender de ataques de flood: a


alteração do Default user level. Funciona mais ou menos assim : em geral, o script considera que todos


os usuários comuns têm o nível 2. Quando ele recebe algum CTCP de um usuário nível 2, ele responde


normalmente e muda esse nível default para 50, por 12 segundos. Se durante esses 12 segundos ele


receber um CTCP de um usuário nível 50, ele responde automaticamente, e eleva o nível padrão para


300. Se ele receber um CTCP de um usuário nível 300, ele passa a ignorar automaticamente todos os


CTCPs deste usuário por 180 segundos, e eleva o nível padrão para 400. Finalmente, se ele receber


algum CTCP de usuários nível 400, ele não faz absolutam,ente nada : nem responde, nem ignora :)





    Eis o script : 








    #protecao on


    ctcp 2:* {


      .dlevel 50


      .timer 1 15 /.dlevel 2


    }


    ctcp 50:* {


      .dlevel 300 :!


    }


    ctcp 300:* {


      .dlevel 400


      ignore -tu180 $nick 2


      .notice $nick FLOOD Ignorado *!* $+ $site $+ . Espere 180 segundos..


    }


    ctcp 400:*:halt :!


    #Protecao end














                     PROTEÇÃO DE CANAL








    A proteção de canal é completamente diferente da proteção de usuário. Seu objetivo é retirar dos


canais usuários que perturbam a ordem. Existem muitos tipos de proteção de canal : badword kick,


shutup kick, flood kick são os mais comuns. Veremos aqui um exemplo de rotina de flood kick, retirada


do DusK 5.0.





    O funcionamento desta rotina é MUITO simples : ao receber um texto, action ou notice de um


usuário nível 2 (padrão), o script muda o nível deste usuário para 20 por 4 segundos. Se receber algo


de um usuário nível 20, o script simplesmente eleva o nível deste usuário para 21. E se finalmente


receber um texto de um usuário nível 21, o script kicka (se a vítima não for op do canal) ou adverte (no


caso de a vítima ser um op). 








    #canal3 off


    ON @2:ACTION:*:#:.auser -a 20 $nick | .timer 1 4 .ruser 20,21 $nick


    ON +20:ACTION:*:#:.ruser 20 $nick | .auser -a 21 $nick


    ON +21:ACTION:*:# {


      if ($nick isop $chan) { .ruser 21 $nick | .notice $nick %AutoNotice }


      else { .ruser 21 $nick | .kick $chan $nick %AutoKick }


    }


    ON @2:TEXT:*:#:.auser -a 20 $nick | .timer 1 4 .ruser 20,21 $nick


    ON +20:TEXT:*:#:.ruser 20 $nick | .auser -a 21 $nick


    ON +21:TEXT:*:# {


      if ($nick isop $chan) { .ruser 21 $nick | .notice $nick %AutoNotice }


      else { .ruser 21 $nick | .kick $chan $nick %AutoKick }


    }


    ON @2:NOTICE:*:#:.auser -a 20 $nick | .timer 1 4 .ruser 20,21 $nick


    ON +20:NOTICE:*:#:.ruser 20 $nick | .auser -a 21 $nick


    ON +21:NOTICE:*:# {


      if ($nick isop $chan) { .ruser 21 $nick | .notice $nick %AutoNotice }


      else { .ruser 21 $nick | .kick $chan $nick %AutoKick }


    }


    #canal3 end





    Este mesmo mecanismo apresentado pode ser facilmente convertido para kickar por join flood,


topic flood, op/deop flood (mass deop), kick flood (mass kick), etc. 


